

Module 19

Representatie van signalen

Onderwerp	Fourier-transformatie van continue en discrete signalen; Bemonsteringsstelling van Shannon, Heaviside-functie
Voorkennis	Fourier-transformatie
Expressies	<code>fourier</code> , <code>Heaviside</code> , <code>listplot</code>
Bibliotheken	<code>inttrans</code>
Zie ook	Module 8.

19.1 Inleiding⁴⁷

In deze module kijken we naar een aantal verschillende manieren om een signaal te beschrijven. Een signaal is in deze context een éédimensionale functie (reëel- of complexwaardig) van één variabele, die in veel gevallen de tijd voorstelt.

Achtereenvolgens komen de volgende onderwerpen aan de orde:

- (1) Signalen in continue tijd (functies van \mathbb{R} naar \mathbb{R} of \mathbb{C});
- (2) Stukjes signaal in continue tijd (functies van een (eindig) interval van \mathbb{R} naar \mathbb{R} of \mathbb{C});
- (3) Periodieke signalen;
- (4) Signalen in discrete tijd (functies van \mathbb{Z} naar \mathbb{R} of \mathbb{C}). Deze signalen ontstaan vaak door bemonstering (sampling) van een signaal in continue tijd;
- (5) Stukjes signaal in discrete tijd (functies van een (eindige) aaneengesloten deelverzameling van \mathbb{Z} naar \mathbb{R} of \mathbb{C});

Bij elk van deze onderwerpen geven we de bijbehorende representatie in het frequentiedomein. Hierbij speelt de Fourier-transformatie in enige vorm een belangrijke rol. We zullen ook ingaan op het verband tussen een signaal in continue tijd en het discrete signaal dat ontstaat door bemonstering van dit signaal.

In het vervolg zullen we signalen in het tijddomein noteren met een kleine letter (bijvoorbeeld $f(t)$) en de bijbehorende signalen in het frequentiedomein met de corresponderende hoofdletter (bijvoorbeeld $F(\omega)$).

⁴⁷Deze module is grotendeels gebaseerd op het boek: G. Galati (Ed.), *Advanced Radar Techniques and Systems*, Peregrinus (IEE) 1993.

19.2 Signalen in continue tijd

Een signaal in continue tijd is een functie f van een (eindig of oneindig) interval naar \mathbb{R} of \mathbb{C} . Wanneer f absoluut integreerbaar is, dat wil zeggen dat de integraal $\int_{-\infty}^{\infty} |f(t)| dt < \infty$, wordt zijn Fourier-getransformeerde $F(\omega)$ gegeven door

$$F(\omega) = \int_{-\infty}^{\infty} f(t) e^{-j\omega t} dt. \quad (19.1)$$

De inverse Fourier-getransformeerde van $F(\omega)$ is

$$f(t) = \frac{1}{2\pi} \int_{-\infty}^{\infty} F(\omega) e^{j\omega t} d\omega. \quad (19.2)$$

De verzameling functiewaarden $F(\omega)$ noemen we ook wel het spectrum van f .

19.3 Fourier-transformatie

`fourier`

`invfourier`

De Fourier-getransformeerde $F(\omega)$ van een functie $f(t)$ kan worden berekend met de procedure `fourier`, uit de bibliotheek `inttrans`. Deze bibliotheek bevat ook een procedure voor de inverse Fourier-transformatie (`invfourier`) en daarnaast ook procedures voor de Laplace-transformatie en voor nog enkele andere integraaltransformaties.

Voorbeeldopgave

Gegeven een functie `recta(t)` met

$$\text{rect}_a(t) = \begin{cases} 1/a, & \text{als } -a/2 < t < a/2; \\ 0 & \text{elders.} \end{cases} \quad (19.3)$$

Bereken voor $a = 1$, $a = 2$ en $a = 4$ de Fourier-getransformeerde van f . Maak de verschillen zichtbaar in een grafische voorstelling.

Voorbeeldsessie

```
> recta := piecewise( t<-a/2, 0, t<a/2, 1/a, 0 );
```


$$\text{rect}_a := \begin{cases} 0 & t < -\frac{a}{2} \\ \frac{1}{a} & t < \frac{a}{2} \\ 0 & \text{otherwise} \end{cases}$$

```
> rect := unapply(recta,a,t):
> f1:=rect(1,t);
```

$$f1 := \begin{cases} 0 & t < -\frac{1}{2} \\ 1 & t < \frac{1}{2} \\ 0 & \text{otherwise} \end{cases}$$

```
> f2:=rect(2,t):    f3:=rect(4,t):
> with(inttrans):
> F1:=fourier(f1,t,omega);
F1 := \frac{2 \sin(\frac{\omega}{2})}{\omega}
> F2 := fourier(f2,t,omega):
F3 := fourier(f3,t,omega):
> plot( [abs(F1),abs(F2),abs(F3)], omega=-5*Pi..5*Pi,
        color=[red,green,black] ):
```

(zie figuur 23)



FIGUUR 23. Fourier-getransformeerden van blokfuncties

Toelichting

In figuur 23 is te zien dat $F(\omega)$ een smallere piek krijgt naarmate $\text{rect}_a(t)$ breder wordt; de ‘buitenste’ grafiek hoort bij de kleinste waarde van a . De *breedte* van de piek in het frequentiedomein is ruwweg omgekeerd evenredig met de breedte van het signaal in het tijddomein. De *hoogte* van de eerste piek naast het maximum (in het frequentiedomein) is in dit geval steeds hetzelfde. \diamond

19.4 Heaviside-functies

Voor stuksgewijs gedefinieerde functies, zoals $\text{rect}_a(t)$ in het voorbeeld van §19.3 wordt vaak gebruik gemaakt van de *eenheidsstap-functie* of *Heaviside-functie*. Deze is gedefinieerd als:

$$H(t) = \begin{cases} 1 & \text{als } t > 0 \\ 0 & \text{als } t < 0, \end{cases}$$

en is dus niet gedefinieerd als $t = 0$. (In sommige boeken komt men ook wel $H(0) = \frac{1}{2}$ tegen.) Bijvoorbeeld de functie $\text{rect}_a(t)$ (19.3) kan op verschillende manieren met behulp van Heavisidefuncties worden geschreven:

$$\text{rect}_a(t) = \frac{1}{a} H(a + 2t) H(a - 2t) = \frac{1}{a} (H(2t + a) - H(2t - a)).$$

Het product $H(a + 2t) H(a - 2t)$ is volgens bovenstaande definitie alleen gelijk aan 1 als $a + 2t$ en $a - 2t$ beide > 0 zijn, en in alle overige gevallen gelijk aan 0. Ga zelf na dat de andere mogelijkheid óók tot het gewenste resultaat leidt.

Heaviside

De mapleprocedure `Heaviside` kan zo worden gebruikt om allerlei *stapfuncties* mee te definiëren. Een *piecewise*-functie kan met een `convert`-commando worden herschreven tot een uitdrukking met Heaviside-functies.

19.5 Stukjes signaal in continue tijd

In de praktijk kunnen we maar een eindig stukje van een signaal $f(t)$ bestuderen, zeg, voor $t \in [a, b]$. We kunnen dit weergeven door vermenigvuldiging van f met een zogenoemde weegfunctie w , waarbij $w(t)$ klein is voor $t \notin [a, b]$. De precieze keuze voor w hangt sterk af van het type analyse dat men wenst te doen op het signaal $f(t)$. In het frequentiedomein wordt de weging een *convolutie*:

$$f(t)w(t) \quad \text{correspondeert met} \quad \frac{1}{2\pi} F(\omega) * W(\omega),$$

met

$$F(\omega) * W(\omega) = \int_{-\infty}^{\infty} F(\omega - s)W(s) ds.$$


In de volgende voorbeeldopgave zullen we het effect van de keuze van w illustreren.

Voorbeeldopgave

Gegeven zijn $f(t) = e^{-|t|}$, voor $t \in \mathbb{R}$, en twee weegfuncties namelijk $w_1(t) = \text{rect}_\pi(t)$, en $w_2(t) = \cos^2 t$ voor $|t| \leq \frac{\pi}{2}$, en 0 elders.

Teken in één figuur $f(t)$, $f(t)w_1(t)$ en $f(t)w_2(t)$.

Bereken de Fourier-getransformeerden van deze drie functies, en maak de verschillen zichtbaar in een plaatje.



FIGUUR 24. De functie $f(t) = e^{-|t|}$, gewogen met $w_1(t) = \text{rect}_\pi(t)$ (dikke lijn), en $w_2(t) = \cos^2 t$ voor $|t| \leq \frac{\pi}{2}$, en 0 elders (streepjeslijn)

Voorbeeldsessie

```
> restart; with(inttrans): with(plots):
> f := t -> exp(-abs(t));
                                f := t -> e^{(-|t|)}
> w1 := (Heaviside(t+Pi/2) - Heaviside(t-Pi/2)):
w2 := w1*(cos(t))^2:
> p1 := plot(f(t),t=-Pi..Pi, linestyle=DOT, color=black):
p2 := plot(f(t)*w1,t=-Pi..Pi, color=black, thickness=4):
p3 := plot(f(t)*w2,t=-Pi..Pi,linestyle=DASH, color=black):
> display({p1,p2,p3});
(zie figuur 24)
> F1 := unapply( simplify(fourier(f(t),t,omega)), omega );
                                F1 := \omega \to \frac{2}{1 + \omega^2}
> W1 := unapply( simplify(fourier(w1,t,omega)), omega );
                                W1 := \omega \to \frac{2 \sin(\frac{1}{2} \pi \omega)}{\omega}
> W2:=unapply( simplify(fourier(w2,t,omega)), omega );
```

$$W2 := \omega \rightarrow -\frac{4 \sin\left(\frac{1}{2} \pi \omega\right)}{\omega(\omega^2 - 4)}$$

```
> int(F1(s)*W1(omega-s),s=-infinity..infinity)/(2*Pi):
F1W1 := unapply( expand(%), omega );
```

$$F1W1 := \omega \rightarrow \frac{2}{1 + \omega^2} - \frac{2 \omega \sin\left(\frac{1}{2} \pi \omega\right) \sinh\left(\frac{\pi}{2}\right)}{1 + \omega^2} - \frac{2 \cos\left(\frac{1}{2} \pi \omega\right) \cosh\left(\frac{\pi}{2}\right)}{1 + \omega^2} + \frac{2 \omega \sin\left(\frac{1}{2} \pi \omega\right) \cosh\left(\frac{\pi}{2}\right)}{1 + \omega^2} + \frac{2 \cos\left(\frac{1}{2} \pi \omega\right) \sinh\left(\frac{\pi}{2}\right)}{1 + \omega^2}$$

```
> 1/(2*Pi)*int(F1(s)*W2(omega-s),s=-infinity..infinity):
F1W2 := unapply( simplify(expand(%)), omega );
```

$$F1W2 := \omega \rightarrow 2(15 - 2 \omega^3 \sin\left(\frac{1}{2} \pi \omega\right) \cosh\left(\frac{\pi}{2}\right) - 10 \cos\left(\frac{1}{2} \pi \omega\right) \cosh\left(\frac{\pi}{2}\right) + \omega^4 + 14 \omega \sin\left(\frac{1}{2} \pi \omega\right) \cosh\left(\frac{\pi}{2}\right) + 2 \omega^3 \sin\left(\frac{1}{2} \pi \omega\right) \sinh\left(\frac{\pi}{2}\right) - 14 \omega \sin\left(\frac{1}{2} \pi \omega\right) \sinh\left(\frac{\pi}{2}\right) + 10 \cos\left(\frac{1}{2} \pi \omega\right) \sinh\left(\frac{\pi}{2}\right) - 6 \omega^2 \cos\left(\frac{1}{2} \pi \omega\right) \sinh\left(\frac{\pi}{2}\right) + 6 \omega^2 \cos\left(\frac{1}{2} \pi \omega\right) \cosh\left(\frac{\pi}{2}\right)) / ((5 + 4 \omega + \omega^2)(5 - 4 \omega + \omega^2)(1 + \omega^2))$$

```
> p4 := plot( abs(F1W1(omega)), omega=-3*Pi..3*Pi, color=black):
p5 := plot( abs(F1W2(omega)), omega=-3*Pi..3*Pi, color=black,
linestyle=DASHDOT ):
p6 := plot( abs(F1(omega)), omega=-3*Pi..3*Pi, color=black,
linestyle=DASH):
```

```
> display({p4,p5,p6});
```

(zie figuur 25, links) De piek van de functies bij 0 is niet overal gelijk. Wanneer we dit willen bereiken voor een betere analyse van deze figuur moeten we hiervoor corrigeren:

```
> fw1piek := evalf(Int(f(t)*w1,t=-infinity..infinity));
```

$$fw1piek := 1.5842408470$$

```
> fw2piek := evalf(Int(f(t)*w2,t=-infinity..infinity));
```

$$fw2piek := 1.0336963390$$

```
> fpiek := evalf(Int(f(t),t=-infinity..infinity));
```

$$fpiek := 2.0000000000$$


```
> p4 := plot( abs(fpiek/fw1piek*F1W1(omega)), omega=-3*Pi..3*Pi,
color=black):
```

```
p5 := plot( abs(fpiek/fw2piek*F1W2(omega)), omega=-3*Pi..3*Pi,
color=black, linestyle=DASHDOT):
```

```
p6 := plot( abs(F1(omega)), omega=-3*Pi..3*Pi,
color=black, linestyle=DASH):
```

```
> display({p4,p5,p6});
```

(zie figuur 25, rechts)



FIGUUR 25. Fourier-getransformeerden van de functie $f(t) = e^{-|t|}$, gewogen met $w_1(t) = \mathbf{rect}_\pi(t)$, en met $w_2(t) = \cos^2 t$ voor $|t| \leq \frac{\pi}{2}$, en 0 elders. Links ongenormeerd en rechts genormeerd

Toelichting

We hebben de Fourier-getransformeerden van fw_1 en fw_2 kunnen berekenen door gebruik te maken van de convolutie. In het resultaat zien we het effect van de weegfunctie terug. Zowel bij w_1 als bij w_2 is de piek in het frequentiedomein verbreed. Dit komt door de eindigheid van de drager van de weegfunctie. Bij gebruik van de rechthoeksfunctie zien we de $\sin(\omega)/\omega$ amplitudeverdeling terug, zoals correspondeert met de Fourier-getransformeerde van een rechthoeksfunctie.

De hoogtes van de centrale pieken zijn verschillend, en worden dus kennelijk door de weegfunctie beïnvloed. Dit is te begrijpen door te kijken naar de formule voor de Fourier-transformatie, en in die formule $\omega = 0$ in te vullen. Zo is

$$\begin{aligned} \frac{1}{2\pi}(F * W_1)(\omega = 0) &= \int_{-\infty}^{\infty} f(t)w_1(t) dt \approx 1.58, \\ \frac{1}{2\pi}(F * W_2)(\omega = 0) &= \int_{-\infty}^{\infty} f(t)w_2(t) dt \approx 1.03, \\ \frac{1}{2\pi}(F)(\omega = 0) &= \int_{-\infty}^{\infty} f(t) dt \approx 2. \end{aligned}$$

Wanneer we normeren ontstaat een netter plaatje, waarin we de resultaten goed kunnen vergelijken. \diamond

19.6 Periodieke signalen in continue tijd

Een periodiek signaal is een signaal f waarvoor geldt $f(t+T) = f(t)$ voor alle t , en zekere T , de periode. Voor periodieke signalen bestaat de integraal (19.1) niet. Wel kunnen we f schrijven als een reeks, de *Fourier-reeks* van periodieke functies:

Als f periodiek is met periode T , en stuksgewijs glad op $[-\frac{T}{2}, \frac{T}{2}]$, dan geldt

$$\frac{f(t_+) + f(t_-)}{2} = \sum_{k=-\infty}^{\infty} f_k e^{j\omega_T k t},$$

met

$$\omega_T = \frac{2\pi}{T}, \quad f(t_+) = \lim_{s \downarrow t} f(s), \quad f(t_-) = \lim_{s \uparrow t} f(s),$$

en waarbij

$$f_k = \frac{1}{T} \int_{-T/2}^{T/2} f(t) e^{-jk\omega_T t} dt.$$

De set coëfficiënten f_k kan worden opgevat als een discreet *spectrum* van f , gedefinieerd op de discrete frequenties die een veelvoud zijn van $\omega_T = \frac{2\pi}{T}$. Dit spectrum wordt ook wel aangeduid met de term *lijnspectrum*.

Voorbeeldopgave

Gegeven de periodieke functie f , met $f(t) = 1$ voor $|t| \leq \frac{1}{2}$, $f(t) = 0$ voor $\frac{1}{2} < |t| < 1$, en verder periodiek voortgezet met periode $T = 2$. Bereken de Fourier-reeks van f en teken de benadering van f door respectievelijk 1, 5 en 11 termen mee te nemen.

Voorbeeldsessie

```
> restart;
> f := t -> Heaviside(t+1/2)-Heaviside(t-1/2);
      f := t -> Heaviside(t + 1/2) - Heaviside(t - 1/2)
> T := 2: omega[0] := 2*Pi/T:
> fk := k -> 1/T*int(f(t)*exp(-I*k*omega[0]*t), t=-1/2..1/2);
```

$$fk := k \rightarrow \frac{1}{T} \int_{-1/2}^{1/2} f(t) e^{(-I k \omega_0 t)} dt$$

```
> fseries1 := fk(0);
```

$$fseries1 := \frac{1}{2}$$


```


> add( fk(k)*exp(I*k*omega[0]*t), k=-5..5 ):
fseries2 := simplify(%);

      fseries2 :=  $\frac{1}{30} \frac{-20 \cos(3 \pi t) + 12 \cos(5 \pi t) + 60 \cos(\pi t) + 15 \pi}{\pi}$ 
> add( fk(k)*exp(I*k*omega[0]*t), k=-11..11 ):
fseries3 := simplify(%);

      fseries3 :=  $\frac{1}{6930} (-4620 \cos(3 \pi t) + 2772 \cos(5 \pi t) + 1540 \cos(9 \pi t)$ 
       $- 1980 \cos(7 \pi t) + 13860 \cos(\pi t) - 1260 \cos(11 \pi t) + 3465 \pi) / \pi$ 
> plot( [fseries1,fseries2,fseries3], t=-1..1,
      color=[black,red,blue],
      tickmarks=[[ -1,0,1], [0.2,0.4,0.6,0.8,1]] );

```

(zie figuur 26)



FIGUUR 26. Afgebroken Fourier-reeksen van een blokfunctie

Toelichting

We zien dat de Fourier-reeks reëel is; dit komt omdat f even is. We zien bovendien dat Maple niet uit zichzelf de makkelijkste representatie kiest. Daarom geven we eerst een `simplify` en gaan pas daarna de grafiek tekenen. Dit spaart rekentijd. Naast `simplify` wil `evalc` ook weleens helpen. \diamond

19.7 Signalen in discrete tijd

Een signaal in discrete tijd is een signaal waarvan de functiewaarden slechts bekend zijn op een discrete verzameling tijdstippen. Wij beschouwen hier alleen tijdstippen die een geheel veelvoud zijn van een getal T_s . Hiervoor gebruiken we de notatie

$$f_s[n] = f(nT_s).$$

Vaak worden de waarden $f_s[n]$ verkregen door *bemonstering* ('sampling') van een signaal in continue tijd. In dit geval noemen we T_s de samplertijd. We zullen functies die verkregen zijn uit een bemonsteringsproces voorzien van een subscript s (in het tijddomein) of S (in het frequentiedomein).

We hebben in het geval van een periodieke functie gezien dat het spectrum discreet is. Omgekeerd blijkt dat de representatie van een signaal in discrete tijd correspondeert met een periodieke functie in het frequentiedomein:

$$F_S(\omega) = T_s \sum_{n=-\infty}^{\infty} f_s[n] e^{-jn\omega T_s}. \quad (19.4)$$

Duidelijk is dat F_S periodiek is met periode $\frac{2\pi}{T_s}$. De functie $F_S(\omega)$ kan benaderd worden door slechts een eindig aantal termen te gebruiken:

$$F_S(\omega) \approx T_s \sum_{n=-N}^N f_s[n] e^{-jn\omega T_s}. \quad (19.5)$$

Voorbeeldopgave

Gegeven de rij monsters $f_s[n] = e^{-(nT_s)^2} (\cos(2\pi nT_s) + 2 \cos(5\pi nT_s))$. Kies achtereenvolgens $T_s = 0.1$, $T_s = 0.2$, $T_s = 0.3$ en $T_s = 0.6$, en teken voor elk geval in een plaatje één periode van $F_S(\omega)$.

Voorbeeldsessie


```
> restart; with(inttrans): with(plots):
> f := t -> exp(-t^2)*(cos(2*Pi*t)+2*cos(5*Pi*t));
      f := t -> e(-t2) (cos(2 π t) + 2 cos(5 π t))
> F := unapply( simplify(fourier(f(t),t,omega)), omega );
      F := ω -> 1/2 √π (2 e(-1/4(ω+5π)2) + e(-1/4(ω+2π)2)
      + 2 e(-1/4(ω-5π)2) + e(-1/4(ω-2π)2))
> FS := (omega,T,N) ->
      evalc( sum(T*f(n*T)*exp(-I*n*omega*T), n=-N..N) );
      FS := (ω, T, N) -> evalc( ∑n=-NN T f(n T) e(-I n ω T))
> p1 := plot( FS(omega,0.1,5), omega=-10*Pi..10*Pi, color=black ):
      t1 := textplot( [21, Re(FS(21.,0.1,5)), " N=5"],
      align={RIGHT,ABOVE} ):
> p2 := plot( FS(omega,0.1,11), omega=-10*Pi..10*Pi, color=blue ):
> p3 := plot( FS(omega,0.1,17), omega=-10*Pi..10*Pi, color=green ):
      t3 := textplot( [16.2, Re(FS(16.2,0.1,17)), " N=17"],
      align={RIGHT,ABOVE} );
```

```

> pFpoints:= pointplot( {seq( [n,F(n)], n=-30..30,0.5 ) } ):
> display({pFpoints,p1,t1,p2,p3,t3}, title="T=0.1",
         tickmarks=[[ -5*Pi=typeset(-5*Pi), -2*Pi=typeset(-2*Pi),
                    -5*Pi=typeset(-5*Pi), -2*Pi=typeset(-2*Pi)],
                    spacing(0.5)], axes=framed );
(zie figuur 27, links)
> p4 := plot( FS(omega,0.1,17), omega=-10*Pi..10*Pi, color=green,
             thickness=4 ):
> p5 := plot( FS(omega,0.2,17), omega=-10*Pi..10*Pi, color=blue ):
t5 := textplot( [16.5,Re(FS(16.5,0.2,17))," T=0.2"],
               align={RIGHT,ABOVE} ):
> p6 := plot( FS(omega,0.3,17), omega=-10*Pi..10*Pi, color=black ):
t6 := textplot( [28,Re(FS(28,0.3,17))," T=0.3"],
               align={RIGHT,ABOVE} ):
> display({p4,p5,t5,p6,t6}, title="N=17",
         tickmarks=[[ -5*Pi=typeset(-5*Pi), -2*Pi=typeset(-2*Pi),
                    -5*Pi=typeset(-5*Pi), -2*Pi=typeset(-2*Pi)], [0,1,2,3]],
         axes=framed );

```

(zie figuur 27, rechts)



FIGUUR 27. Spectrum van het continue signaal (links) en van het bemonsterde signaal (rechts)

Toelichting

We controleren eerst of het spectrum er ongeveer uitziet zoals we verwachten, en hoeveel termen we moeten meenemen. In dit geval liggen de pieken ongeveer daar waar we ze verwachten, namelijk rond $\omega = \pm 2\pi$ en $\omega = \pm 5\pi$. We zijn tevreden met 35 termen, dus $n = -17 \dots 17$. Wanneer we de bemonsteringsfrequentie vermindere, zien we dat de pieken rond $\pm 5\pi$ verdwijnen. Voor $T_s = 0.2$ kunnen we ze nog juist beide zien, maar voor $T_s = 0.3$ zijn de pieken met de hoogste frequentie verdwenen. Wanneer $T_s > 0.5$ verdwijnt ook de piek rond $\omega = \pm 2\pi$. \diamond

Het voorbeeld geeft aan dat we niet zomaar een willekeurige bemonsteringsfrequentie kunnen kiezen: deze frequentie moet hoog genoeg zijn. We hebben nu de Fourier-paren $f(t) \leftrightarrow F(\omega)$, en $f_s[n] \leftrightarrow F_S(\omega)$. De functie $F_S(\omega)$ heeft een relatie met de Fourier-getransformeerde $F(\omega)$ van $f(t)$. De functie $F_S(\omega)$ bestaat uit een oneindige som van replica's van $F(\omega)$, elk verschoven over een afstand $\omega_s = \frac{2\pi}{T_s}$, de periode van $F_S(\omega)$:

$$F_S(\omega) = \sum_{n=-\infty}^{\infty} F(\omega - n\omega_s).$$

In het geval dat $F(\omega) = 0$ voor $|\omega| > \omega_b$, voor zekere ω_b en $\omega_s > 2\omega_b$ overlappen de replica's niet. De frequentie ω_b heet de *bandbreedte* van f . In zo'n geval bezit één willekeurige replica alle informatie over $F_S(\omega)$, en dus ook van $F(\omega)$ en $f(t)$. Er is dan dus geen informatie verloren gegaan door sampling van $f(t)$. De frequentie $\omega_s = 2\omega_b$ noemt men de *Nyquist-frequentie*; het resultaat dat we hier hebben beschreven is het onderwerp van de stelling van Shannon.

Wanneer $\omega_s > 2\omega_b$, kunnen we $f(t)$ reconstrueren uit de samples $f[n]$, volgens

$$f(t) = \sum_{n=-\infty}^{\infty} f_s[n] \frac{\sin(\omega_s(t - nT_s)/2)}{\omega_s(t - nT_s)/2}. \quad (19.6)$$

In de volgende voorbeeldopgave wordt een en ander geïllustreerd.

Voorbeeldopgave


Gegeven het signaal

$$f(t) = e^{-\frac{5}{2}t^2} (54 - 625t^2 + 625t^4).$$

Bereken, volgens (19.4) de benaderingen van het spectrum door f te bemonsteren met $T_s = \pi/10$ en $\pi/14$. Teken deze benaderingen op het interval $-36 < \omega < 36$. Wat kunt u zeggen over de bemonsteringsfrequentie die zou moeten worden gebruikt? Verifieer uw antwoord door de Fourier-getransformeerde van f te bepalen en in een plaatje te tekenen. Wat is de hoogste frequentie die in het signaal voorkomt?

Voorbeeldsessie

```
> restart; with(inttrans):
> f := t -> exp(-5/2*t^2)*(54-625*t^2+625*t^4);
      f := t -> e(-5/2 t2) (54 - 625 t2 + 625 t4)
> plot( f(t), t=-4.5..4.5, color=black );
(zie figuur 28)
> Fn := (N,omega,T) -> sum( T*f(n*T)*exp(-I*n*omega*T), n=-N..N):
```



FIGUUR 28. De functie $f(t) = e^{-\frac{5}{2}t^2} (54 - 625t^2 + 625t^4)$


```

> T1 := Pi/10: T2 := Pi/14:
> F1 := evalc(Fn(15,omega,T1)): F2 := evalc(Fn(15,omega,T2)):
> plot( F1, omega=-36..36, numpoints=250, title=typeset(T=Pi/10) );
(zie figuur 29, links)
> plot( F2, omega=-36..36, numpoints=250, title=typeset(T=Pi/14) );
(zie figuur 29, rechts)
> Fomega := evalc(fourier(f(t),t,omega)):
  plot( Fomega, omega=-36..36, numpoints=250,
        title="De Fouriergetransformeerde van f" );
(zie figuur 30, links)
> omega_s1 := 2*Pi/T1: omega_s2 := 2*Pi/T2:
> ft1 := evalf( sum(
  f(n*T1)*sin(omega_s1*(t-n*T1)/2)/(omega_s1*(t-n*T1)/2),
  n=-15..15) ):
  p1 := plot( ft1-f(t), t=-4.5..4.5, color=black ):
> ft2 := evalf( sum(
  f(n*T2)*sin(omega_s2*(t-n*T2)/2)/(omega_s2*(t-n*T2)/2),
  n=-15..15) ):
  p2 := plot( 1000*(ft2-f(t)), t=-4.5..4.5, color=black, thickness=3 ):
> plots:-display({p1,p2}, axes=framed);
(zie figuur 30, rechts)


```

Toelichting

We zien hier geïllustreerd dat de benadering van de Fourier-getransformeerde die verkregen wordt via het sampleproces een periodieke functie is, met periode gelijk aan $\frac{2\pi}{T_s}$. Uit de grafieken van de benaderingen van het spectrum (figuur 29) is te zien dat vermoedelijk de bemonsteringsfrequentie groter moet zijn dan $\pi/10$; de frequentie $\pi/14$ is wat te groot. Na berekening van de Fourier-getransformeerde van f blijkt, uit de grafiek hiervan (figuur 30, links), dat de maximale frequentie die voorkomt ergens in de buurt van $\omega = 12$ ligt.



FIGUUR 29. Spectrum van het bemonsterde signaal



FIGUUR 30. Spectrum van het continue signaal (links); reconstructiefout (rechts)

We zien dat de reconstructie van f uit de samples $f[n]$ veel beter is als $\omega_s > 2\omega_b$. In de grafieken (figuur 30, rechts) zijn de reconstructiefouten zichtbaar gemaakt. Merk op dat we de fout corresponderend met $\omega > 2\omega_b$ met een factor 1000 hebben vermenigvuldigd! \diamond

19.8 Stukje signaal in discrete tijd

Net als in het continue geval kunnen we het stukje signaal opvatten als de vermenigvuldiging van het oorspronkelijke signaal met een window-functie. De (on)hebbelijkheden van de window-functie zijn ook in het discrete geval aanwezig, zoals blijkt uit de volgende beeldsessie.

Voorbeeldopgave

Gegeven zijn $f(t) = e^{-|t|}$, voor $t \in \mathbb{R}$, en twee windowfuncties $w_1(t) = \text{rect}_\pi(t)$, en $w_2(t) = \cos^2 t$ voor $|t| \leq \frac{\pi}{2}$, en 0 elders. Kies een sampletijd $T_s = 0.1$, en bereken

$$F_S(\omega) = T_s \sum_{n=-N}^N w_I[n] f[n] e^{-jn\omega T_s},$$

voor beide functies $w_1(t)$ en $w_2(t)$.


Voorbeeldsessie

```
> f := t -> exp(-abs(t));
                                     f := t -> e(-|t|)
> w1 := t-> Heaviside(t+Pi/2) - Heaviside(t-Pi/2):
w2 := t-> w1(t)*(cos(t))^2:
> FS1 := (omega,Ts,N) -> Ts*evalc(
    sum(w1(n*Ts)*f(n*Ts)*exp(-I*n*omega*Ts), n=-N..N) );
                                     FS1 := (omega, Ts, N) -> Ts evalc(
                                     sumn=-NN w1(n Ts) f(n Ts) e(-I n omega Ts))
> FS2 := (omega,Ts,N) -> Ts*evalc(
    sum(w2(n*Ts)*f(n*Ts)*exp(-I*n*omega*Ts),n=-N..N) );
Net als in het continue geval normeren we de functies, nu op piek=1:
> FW1piek := evalf( FS1(0,0.1,25) );
                                     FW1piek := 1.577347263
> FW2piek := evalf( FS2(0,0.1,25) );
                                     FW2piek := 1.035370253
> Fpiek := evalf( 0.1*add(f(0.1*n), n=-25..25) );
                                     Fpiek := 1.845568106
> p1 := plot( FS1(omega,0.1,25)/FW1piek,
    omega=-2*Pi/0.1..2*Pi/0.1, color=black ):
p2 := plot( FS2(omega,0.1,25)/FW2piek,
    omega=-2*Pi/0.1..2*Pi/0.1, thickness=3 ):
plots[display]({p1,p2});
```

(zie figuur 31)

Toelichting

De `evalc`-opdracht in de definitie van `FS1` en `FS2` zorgt ervoor dat de functiewaarden $FS1(\omega)$ en $FS2(\omega)$ in de vorm $a + jb$ worden geschreven. Hierdoor kan Maple herkennen dat de expressies reëel zijn.



FIGUUR 31. Reconstructie van het bemonsterde signaal $f(t) = e^{-|t|}$ met twee verschillende vensters

Hier zijn dezelfde fenomenen zichtbaar als in het continue geval. Ook hier normeren we de grafieken. In dit geval is

$$F_S(0) = T_s \sum_{n=-N}^N f_s[n]$$

de piekhoogte. ◇

19.9 Discrete Fourier-transformatie

In de praktijk hebben we niet alleen te maken met slechts een eindig aantal (zeg $2N + 1$) signaalmonsters $f_s[n]$, $n = -N \dots N$. We kunnen bovendien veelal maar een eindig aantal monsters van $F_S(\omega)$ verwerken. Om toch een indruk te krijgen van het spectrum van f , maken we gebruik van (19.5):

$$F_S(\omega) \approx T_s \sum_{n=-N}^N f[n] e^{-jn\omega T_s}.$$

We weten dat $F_S(\omega)$ periodiek is met periode $\omega_s = 2\pi/T_s$. Als we nu het interval $[-\omega_s, \omega_s]$ opdelen in $2N$ gelijke deelintervallen door te kiezen $\omega_k = \omega_s k/N$, $k = -N \dots N$, en noteren $F_S[k] = F_S(\omega_k)$, vinden we

$$F_S[k] = T_s \sum_{n=-N}^N f[n] e^{-j2\pi nk/(2N+1)}.$$

Deze transformatie, die samples $f_s[n]$ in het tijddomein afbeeldt op samples $F_S[k]$ in het frequentiedomein, staat bekend onder de naam

discrete Fourier-transformatie. Omgekeerd kunnen we $f_s[n]$ uitdrukken in termen van $F_S[k]$ door de inverse discrete Fourier-transformatie:

$$f_s[n] = \frac{1}{T_s(2N+1)} \sum_{k=-N}^N F_S[k] e^{2\pi jnk/(2N+1)}.$$

Merk op dat zowel $f[n]$ als $F_S[k]$ periodiek is met periode $2N+1$. Wanneer het aantal samples gelijk is aan een macht van 2, bestaat er een efficiënte implementatie voor de discrete Fourier-transformatie. Deze staat bekend onder de naam *Fast Fourier Transform* (FFT). Wegens de toenemende rekenkracht van processoren wordt het belang van de FFT in vele applicaties minder groot.

Voorbeeldopgave

Gegeven een rij samples van een signaal $f_s[n] = f(nT_s) = \sin(nT_s) + \sin(3nT_s)$ met sample-interval $T_s = 1$. Bepaal de discrete fourier-transformatie van de rij $f_s[n]$. Neem $N = 50$. Teken een grafiek en interpreteer het resultaat.


Voorbeeldsessie

```
> f := t -> sin(t)+sin(3*t);
                                f := t -> sin(t) + sin(3t)
> Ts := 1:    N := 50:
Dit worden de samples (in het tijd-domein) waarmee we verder werken:
> fwaarden := seq( evalf(f(n*Ts)), n=-N..N ):
> fs := Array( -N..N, [fwaarden] ):
In het frequentie-domein:
> Fwaarden := seq( evalf(
    add(fs[n]*exp(-I*2*Pi*n*k/(2*N+1)), n=-N..N)
    ), k=-N..N ):
> FS := Array( -N..N, [Fwaarden] ):
> plot( [seq([n,abs(FS[n])], n=-N..N)] );
```

(zie figuur 32)

Toelichting

We zetten de samples in een Array (zie §8.4) om ze gemakkelijk te kunnen nummeren van $-N$ tot N . De discrete Fourier-getransformeerde $F_S[k]$ hebben we geplot tegen het rangnummer k langs de horizontale as. We zien een grafiek waarbij de horizontale as dus loopt van -50 tot $+50$. Dit interval correspondeert met één periode van het spectrum, namelijk $\frac{2\pi}{T_s}$. De piek rond x -coördinaat 17 correspondeert dus met een frequentie van ongeveer $17 \times \frac{2\pi}{100T_s} \approx 1$; de andere piek



FIGUUR 32. Discrete Fourier-transformatie

correspondeert met een frequentie rond 3. Merk op dat de sample-frequentie van $T_s = 1$ nog juist hoog genoeg is; de Nyquist-frequentie is $\pi/3$. \diamond

Opgave 19.1

Bereken de Fourier-getransformeerden van

- (a) e^{-t^2} ; (b) $\frac{1}{t}$; (c) $\sin(t)$ voor $-1 \leq t \leq 1$.

Opgave 19.2

Gegeven $f(t) = 1/(t^2 + 4)$. Bereken achtereenvolgens de Fourier-getransformeerden van

- (a) $f(t)$; (b) $f(t - a)$; (c) $f(t/b)$; (d) $f((t - a)/b)$.

Gebruik `simplify`, eventueel te combineren met `assume=positive`, en vergelijk de resultaten.

Opgave 19.3

Bepaal de Fourier-reeks van de periodieke functies waarvan hieronder steeds één periode is gegeven:

- (a) $f(t) = \cos^2(t)$ op $[-\pi, \pi]$;
 (b) $f(t) = \sqrt{|t|}$ op $[-1, 1]$;
 (c) $f(t) = \cos(2 \cos(t))$ op $[-\pi/2, \pi/2]$.

Maak een grafiek van het verschil tussen $f(t)$ en de Fourier-reeks, voor verschillende aantallen termen. Waar gaat de convergentie moeilijk, en waarom?

Opgave 19.4

Gegeven de functie $f : t \mapsto \sin(200t)$ op het interval $[0, 2\pi]$.

- Genereer de lijst coördinaatparen $[nT_s, f(nT_s)]$ zodanig dat we een lijst krijgen die geplotted kan worden. Neem $T_s = 0.1$. Verklaar het plaatje. Hoe groot moet T_s wel gekozen worden?
- Wat gebeurt er als je het aan Maple zelf overlaat middels het commando `plot`?

Opgave 19.5

Gegeven de functie f , met $f(x) = x$ voor $|x| < 5$, en $f(x) = 0$ elders. Gegeven verder de monsters $f_s[-5], \dots, f_s[5]$ met $f_s[n] = n$.

- Teken in één plaatje

$$f_s[n] \frac{\sin(\omega_s(t - nT_s)/2)}{\omega_s(t - nT_s)/2}$$

voor een aantal waarden van n .

- Teken in één plaatje op het interval $[-6, 6]$ de functie f_s , alsmede de grafiek van

$$\sum_{n=-5}^5 f[n] \frac{\sin(\omega_s(t - nT_s)/2)}{\omega_s(t - nT_s)/2}.$$

- Probeer een hogere bemonsteringsfrequentie, bijvoorbeeld $T_s = 0.1$. In hoeverre wordt de functie f hierdoor goed benaderd? Waar gaat het fout, en hoe komt dat?

Opgave 19.6

Gegeven de functie $f : t \mapsto \sin(\frac{\pi}{2} \sin(t))$.

- Kies een bemonsteringsinterval $T_s = 0.2$, en bemonster f op het interval $[-4..4]$. We gebruiken dus 41 monsters.
- Benader de Fourier-getransformeerde van f met behulp van de monsters die u bij het vorige onderdeel heeft verkregen.
- Herhaal beide vorige onderdelen voor achtereenvolgens
 - $T_s = 0.4, T_s = 1, N = 41$ (grotere bemonsteringstijd, zelfde aantal monsters; het bemonsteringsinterval wordt dus respectievelijk $[-8, 8]$ en $[-20, 20]$).
 - $N = 41, T_s = 0.1$ en $N = 81, T_s = 0.05$ (zelfde bemonsteringstijd, meer samples).

Verklaar de verschillen.

Opgave 19.7

Gegeven $f : t \mapsto \sin(5t) \text{rect}_{2\pi}(t)$.

- (a) Teken de grafiek van f .
- (b) Bemonster f achtereenvolgens met:
 1. 7 monsters, $T_s = \frac{2\pi}{7}$, op $[-3T_s, 3T_s]$
 2. 11 monsters, $T_s = \frac{2\pi}{10}$, op $[-5T_s, 5T_s]$
 3. 15 monsters, $T_s = \frac{2\pi}{13}$, op $[-7T_s, 7T_s]$

Maak plaatjes waarin f en de drie benaderingen zoals hierboven gegeven zijn gevisualiseerd. Gebruik voor elke grafiek een andere kleur (raadpleeg zonodig `?plots`). Waar is de plot van geval (2)? Hoe komt dat?
- (c) Visualiseer de (gladde) reconstructie van f volgens de formules van Shannon.
- (d) Bepaal de benaderingen van de Fourier-getransformeerde van f , uitgaande van de monsters verkregen bij onderdeel (b). Verklaar het plaatje. Waar komen de pieken (anders dan die bij $\omega = 5$) vandaan? Waarom heeft de benadering uitgaande van onderdeel b2 geen piek bij $\omega = 5$?

Opgave 19.8

Gegeven $f : t \mapsto \sum_{k=-7}^7 \cos(kt)$.

- (a) Bereken $f_s[n] = f(nT_s)$, $n = -N \dots N$, op het interval $[-5/2, 5/2]$, met achtereenvolgens
 1. $N = 3$ (en dus $T = -5/2 \times 1/3$)
 2. $N = 30$ (dus hoe groot is T ?)
 3. $N = 300$ (dus hoe groot is T ?)
- (b) De reconstructies volgens (1) - (3) noemen we $\hat{f}_1, \hat{f}_2, \hat{f}_3$. Maak een plaatje met in een figuur de reconstructies $\hat{f}_1, \hat{f}_2, \hat{f}_3$, en f .
- (c) Teken in verschillende plaatjes de verschilfuncties $\hat{f}_i - f$, voor $i = 1 \dots 3$. Waar komen de benaderingsfouten vandaan?

Opgave 19.9

We gaan in deze opgave na welke effecten optreden bij het bemonsteren van een signaal $f(t)$ met een niet-perfect sampleproces (in de praktijk zijn alle sampleprocessen niet perfect).

- (a) We nemen eerst één sample: definieer $w(\tau, t) = \frac{1}{2\tau} \text{rect}_t(\tau)$. Teken voor verschillende waarden van τ de grafiek van $w(\tau, t)$. Wat kunt u zeggen over de limiet $\lim_{\tau \rightarrow 0} w(\tau, t)$?

- (b) Kunt u al iets zeggen over de relatie tussen het spectrum van w en de waarde van τ ? Teken met behulp van `fourier` het spectrum voor enkele waarden van τ , bijvoorbeeld $\tau \in \{1, 0.1, 0.01\}$.
- (c) Nu gaan we een reeks van deze monsterfuncties construeren ten einde een rij samples van f te kunnen nemen. Definieer in Maple de functie $\text{kam}(T_s, \tau, t, N)$, gegeven door

$$\text{kam}(T_s, \tau, t, N) = \sum_{n=-N}^N w(\tau, t + nT_s).$$

Neem $N = 10$, $T_s = 1/2$, $\tau = 0.05$ en teken de grafiek van kam . Teken ook de Fourier-getransformeerde van kam . Kies, na enig geëxperimenteer, zelf een geschikt interval.

- (d) Neem nu $f(t) = \cos(2t)$. We modelleren het sampleproces door het product $f_s(t) = f(t) \text{kam}(T_s, \tau, t, N)$. Teken een plaatje van f_s voor $T_s = 1/2$, $\tau = 0.05$ en $N = 10$.
- (e) Bereken de Fourier-getransformeerde $F_s(\omega)$ van f_s (door `fourier` toe te passen op f_s). Maak een plaatje van $|F_s(\omega)|$. Neem drie verschillende intervalgroottes: $-15 < \omega < 15$, $-40 < \omega < 40$ en $-400 < \omega < 400$ (dit kan wel even duren!). Neem voldoende veel punten mee (met de optie `numpoints`).
- (f) Als u het goed heeft gedaan, vindt u een grafiek met enkele ‘grote’ lokale maxima en vele ‘kleine’ lokale maxima. Waar komen deze vandaan? Verifieer uw antwoord door te experimenteren met T_s , τ en N .

